

calculus_I

Απλές αλγεβρικές πράξεις

```
5*(x+x)-5*x+2*(x-2)
7*x - 4
```

Προσοχή στον ορισμό των αλγεβρικών μεταβλητών (εκτός από την x που είναι ήδη ορισμένη)

```
y+2*y
Traceback (click to the left of this block for traceback)
...
NameError: name 'y' is not defined
```

```
y=var('y')
```

```
y+2*y
3*y
```

Προσοχή στη χρήση του * για τον πολλαπλασιασμό (είναι απαραίτητο)

```
3y-2
Traceback (click to the left of this block for traceback)
...
SyntaxError: invalid syntax
```

```
3*y-2
3*y - 2
```

```
sin(pi/3)
1/2*sqrt(3)
```

Με το σύμβολο _ αναφερόμαστε στο τελευταίο αποτέλεσμα

```
view(_)
```

$$\frac{1}{2}\sqrt{3}$$

```
(1+2*i)**3
-2*I - 11
```

```
conjugate(1+2*i)
-2*I + 1
```

```
norm(1+2*i)
5
```

```
simplify(arcsin(1/sqrt(2)))
1/4*pi
```

```
simplify(exp(i*pi/6))
1/2*sqrt(3) + 1/2*I
```

```
log(1000,10)
```

```
3
```

```
ln(e)
```

```
1
```

Πηλίκο και υπόλοιπο διαίρεσης ακεραίων

```
7 // 3
```

```
2
```

```
7 % 3
```

```
1
```

Παραγοντοποίηση ακεραίων

```
factor(2^60-1)
```

```
3^2 * 5^2 * 7 * 11 * 13 * 31 * 41 * 61 * 151 * 331 * 1321
```

Απλές Αριθμητικές πράξεις

```
3.25*4.2**2*(1/2+3/8)
```

```
50.1637500000000
```

```
n(sin(pi/10))
```

```
0.309016994374947
```

```
n(pi,700)
```

```
3.14159265358979323846264338327950288419716939937510582097494459230  
816406286208998628034825342117067982148086513282306647093844609550.  
223172535940812848111745028410270193852110555964462294895493038196.  
2881098
```

```
u=2./3.;u^30
```

```
5.21509505084655e-6
```

```
u=2/3;u^30
```

```
1073741824/205891132094649
```

```
n(u^30)
```

```
5.21509505084656e-6
```

```
abs(2+3*i)
```

```
sqrt(13)
```

```
(2+3*i)*(2-3*i)
```

```
13
```

```
(2+3*i)^5
```

```
-597*I + 122
```

Ειδικές συναρτήσεις

```
factorial(50)
```

```
30414093201713378043612608166064768844377641568960512000000000000
```

```
binomial(10,2)
```

```
45
```

Απλοποίηση - Παραγοντοποίηση - Ανάπτυγμα αλγεβρικών εκφράσεων

```
factor(x^14 - 7*x^12 + 21*x^10 - 35*x^8 + 35*x^6 - 21*x^4 + 7*x^2 - 1)
```

```
(x + 1)^7*(x - 1)^7
```

```
f= cos(x)^6 + sin(x)^6 + 3 * sin(x)^2 * cos(x)^2;  
f.simplify_trig()
```

```
1
```

```
f=(1/32*cos(6*x) + 3/16*cos(4*x) + 15/32*cos(2*x) + 5/16)  
f.simplify_trig()
```

```
cos(x)^6
```

```
f = sin(x)^8; f.reduce_trig()
```

```
1/128*cos(8*x) - 1/16*cos(6*x) + 7/32*cos(4*x) - 7/16*cos(2*x) +  
35/128
```

```
factor(x^21 - x^20 - x + 1)
```

```
(x^8 - x^6 + x^4 - x^2 + 1)*(x^4 + x^3 + x^2 + x + 1)*(x^4 - x^3 +  
x^2 - x + 1)*(x^2 + 1)*(x + 1)*(x - 1)^2
```

Επίλυση εξισώσεων - συστημάτων

```
solve(x^4 - 5*x^3 + x^2 + 25*x - 30 == 0, x)
```

```
[x == 3, x == 2, x == -sqrt(5), x == sqrt(5)]
```

```
y,z=var('y,z')
```

```
solve([x+2*x+z==3, 7*x-9*y-2*z==1, x+y+7*z==3], x, y, z)
```

```
[[x == (155/167), y == (94/167), z == (36/167)]]
```

```
solve([x^2+2*y^3==1, 2*x-y==-1], x, y)
```

```
[[x == -0.104602450017335, y == 0.7907950999653299], [x ==  
(-0.728948774892838 + 0.257165370361543*I), y ==  
(-0.4578975497856761 + 0.514330740723086*I)], [x ==  
(-0.728948774892838 - 0.257165370361543*I), y ==  
(-0.4578975497856761 - 0.514330740723086*I)]]
```

```
solve(sin(x+pi/3)==1/2, x)
```

```
[x == -1/6*pi]
```

```
sols=solve(x^4-2*x==0, x);sols
```

```
[x == 1/2*I*sqrt(3)*2^(1/3) - 1/2*2^(1/3), x ==  
-1/2*I*sqrt(3)*2^(1/3) - 1/2*2^(1/3), x == 2^(1/3), x == 0]
```

```
r1=sols[0].rhs();r1
```

```
1/2*I*sqrt(3)*2^(1/3) - 1/2*2^(1/3)
```

```
r2=sols[1].rhs();r2
```

```
-1/2*I*sqrt(3)*2^(1/3) - 1/2*2^(1/3)
```

```
r3=sols[2].rhs();r3
```

```
2^(1/3)
```

```
x1,x2,x3,x4 = [x.rhs() for x in sols]
```

Ορισμός συναρτήσεων

```
f(x)=x*sin(x^2)
```

```
a=var('a')
```

```
f(a)
```

```
a*sin(a^2)
```

```
f(pi)
```

```
pi*sin(pi^2)
```

```
n(f(pi))
```

```
-1.35183114215824
```

```
f
```

```
x |--> x*sin(x^2)
```

Αντικατάσταση σε έκφραση

```
t=var('t')
```

```
o=x^2-2*x*sin(t)
```

```
o.substitute(t=pi/2)
```

```
x^2 - 2*x
```

```
(x^2-2*x^3).substitute(x=1,t=pi/3)
```

```
-1
```

```
eq=x^4-x^2-x-1==0
```

```
solve(eq,x)
```

```
[x == -1/2*(1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3)*(I*sqrt(3) + 1) -  
1/18*(-I*sqrt(3) + 1)/(1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3) + 1/3,  
== -1/2*(1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3)*(-I*sqrt(3) + 1) -  
1/18*(I*sqrt(3) + 1)/(1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3) + 1/3, :  
== (1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3) +  
1/9/(1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3) + 1/3, x == -1]
```

```
p=eq.left();p
```

```
x^4 - x^2 - x - 1
```

```
p.roots()
```

```
[(-1/2*(1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3)*(I*sqrt(3) + 1) -  
1/18*(-I*sqrt(3) + 1)/(1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3) + 1/3,  
1),
```

```
(-1/2*(1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3)*(-I*sqrt(3) + 1) -
1/18*(I*sqrt(3) + 1)/(1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3) + 1/3,
1),
((1/18*sqrt(31)*sqrt(3) + 29/54)^(1/3) + 1/9/(1/18*sqrt(31)*sqrt(3)
+ 29/54)^(1/3) + 1/3,
1),
(-1, 1)]
```

```
s,c=var('s c');f=sin(x)^2/cos(x)^4-tan(x)^2*cos(x)^4;
ff=f.substitute({sin(x): s, cos(x):c, tan(x): s/c});ff
-c^2*s^2 + s^2/c^4
```

Εύρεση πραγματικών ριζών πολυωνύμου αριθμητικά

```
p.roots(ring=RR)
[(-1.0000000000000000, 1), (1.46557123187677, 1)]
```

Εύρεση μιγαδικών ριζών πολυωνύμου αριθμητικά

```
p.roots(ring=CC)
[(-1.0000000000000000, 1),
(1.46557123187677, 1),
(-0.232785615938384 - 0.792551992515448*I, 1),
(-0.232785615938384 + 0.792551992515448*I, 1)]
```

```
sp=p.roots();
z=(x^2-2*x-2).substitute(x=sp[1][0])
```

```
n(z)
-2.10837828597563 - 1.95409339251270*I
```

Λογικές Πράξεις

```
5==3
False
```

```
5>3
True
```

```
8>=2
True
```

```
bool(x==sqrt(x^2))
False
```

Δακτύλιοι μεταβλητών

ZZ: ακέραιοι

QQ: ρητοί

RR: πραγματικοί

```
125 in ZZ
```

True

```
1.5 in ZZ
```

False

```
1.5 in QQ
```

True

```
pi in QQ
```

False

```
pi in RR
```

True

```
1+i in CC
```

True

```
parent(3/5)
```

Rational Field

```
parent(33)
```

Integer Ring

Μπορούμε να θέσουμε προϋποθέσεις σε συμβολικές μεταβλητές

```
assume(x>0);  
bool(x==sqrt(x^2))
```

True

Πρέπει να τις ακυρώσουμε όταν δεν χρειάζονται πια

```
forget(x>0)
```

```
bool(tan(pi/4)==1)
```

True

```
n = var('n'); assume(n, 'integer'); sin(n*pi).simplify()
```

0

```
forget(n, 'integer')
```

Λίστες-Πίνακες

```
s=[17,11,2,3]
```

```
s[0]
```

17

```
s[1]
```

11

Απλός Προγραμματισμός (ανακύκλωση, διακλάδωση)

```
p=x^4-2*x^3-3*x-1;
s=p.roots(ring=CC);s
```

```
[(-0.310319971846122, 1),
 (2.53030083756844, 1),
 (-0.109990432861157 - 1.12314660756467*I, 1),
 (-0.109990432861157 + 1.12314660756467*I, 1)]
```

```
for xx in s:
    print xx[0]
```

```
-0.310319971846122
2.53030083756844
-0.109990432861157 - 1.12314660756467*I
-0.109990432861157 + 1.12314660756467*I
```

```
for i in range(0,len(s)):
    print u'Ρίζα :',i+1,s[i][0]
```


```
Ρίζα : 1 -0.310319971846122
Ρίζα : 2 2.53030083756844
Ρίζα : 3 -0.109990432861157 - 1.12314660756467*I
Ρίζα : 4 -0.109990432861157 + 1.12314660756467*I
```

```
for i in range(0,len(s)):
    root,mult=s[i]
    if imag_part(root)==0:
        print u'Ρίζα',i+1,'(Πραγματική) :',root,', Πολλ:',mult
    else:
        print u'Ρίζα',i+1,'(Μιγαδική) :',root,', Πολλ:',mult
```


```
Ρίζα 1 (Πραγματική): -0.310319971846122 , Πολλ: 1
Ρίζα 2 (Πραγματική): 2.53030083756844 , Πολλ: 1
Ρίζα 3 (Μιγαδική) : -0.109990432861157 - 1.12314660756467*I , Πολλ:
1
Ρίζα 4 (Μιγαδική) : -0.109990432861157 + 1.12314660756467*I , Πολλ:
1
```

Απλά Γραφικά

```
plot(sin(x^3)/(x+1),(x,0,pi))
```



```
x,y=var('x y');
plot3d(x*y*(x^2 - y^2) / (x^2 + y^2), (x,-5,5), (y,-5,5))
```




```
g1=plot(sin(x^3)/(x+1),(x,0,2.5),color='orange');  
g2=plot(sin(x^2),(x,0,2.5),color='blue');  
g1+g2
```

